# $Id: PKGBUILD 174752 2013-01-06 10:43:03Z foutrelis $
# Maintainer: Evangelos Foutras <evangelos@foutrelis.com>
# Contributor: tobias [tobias at archlinux.org]

pkgname=mousepad
pkgver=0.3.0
pkgrel=2
pkgdesc="Simple Text Editor for Xfce4 (based on Gedit)"
arch=('i686' 'x86_64')
url=('http://www.xfce.org/')
license=('GPL2')
groups=('xfce4-goodies')
depends=('gtksourceview2' 'dbus-glib' 'desktop-file-utils')
makedepends=('intltool' 'perl')
options=('!libtool')
install=$pkgname.install
source=(http://archive.xfce.org/src/apps/$pkgname/0.3/$pkgname-$pkgver.tar.bz2)
sha256sums=('10f27506994d0d0b8f9e02555404a144babedab97517abe3b6be8b2d21ff046d')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  ./configure \
    --prefix=/usr \
    --sysconfdir=/etc \
    --libexecdir=/usr/lib/xfce4 \
    --localstatedir=/var \
    --disable-static \
    --disable-debug
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"

  make DESTDIR="$pkgdir" install
}

# vim:set ts=2 sw=2 et:
